```
>
```

Lines and Planes in Space

```
> restart:with(plots):with(plottools):with(VectorCalculus):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
> BasisFormat(false):
```

We first look at the vector equation of the line **P** = **O_P1** + *t***a** = (2, 1, -3) + *t*(2, 5, 2), visualized by the colored arrows, along with the parametrization of the line in **p6**, which is visualized by the black line. In some sense, you can think of the point (2,1,-3), the tip of the red arrow, as 0, with the tips of the other arrows as ±1 or ±2 (direction vectors).

```
> p1:=PlotVector(RootedVector(root=[0,0,0],[2,1,-3]),width=.4,
  head_width=.8,head_length=.8,scaling=constrained,color=red):
  p2:=PlotVector(RootedVector(root=[2,1,-3],[2,5,2]),width=.4,
  head_width=.8,head_length=.8,scaling=constrained,color=blue):
  p3:=PlotVector(RootedVector(root=[4,6,-1],[2,5,2]),width=.4,
  head_width=.8,head_length=.8,scaling=constrained,color=green):
  p4:=PlotVector(RootedVector(root=[2,1,-3],[-2,-5,-2]),width=.4,
  head_width=.8,head_length=.8,scaling=constrained,color=magenta):
  p5:=PlotVector(RootedVector(root=[0,-4,-5],[-2,-5,-2]),width=.4,
  head_width=.8,head_length=.8,scaling=constrained,color=orange):
  p6:=spacecurve([2+2*t,1+5*t,-3+2*t],t=-3..3,color=black):
  display(p1,p2,p3,p4,p5,p6,orientation=[-60,70]);
```



To find the **equation of a plane**, it is sufficient to know a point on the plane **P1** and a normal vector **a** to the plane. We enter such a point and normal vector along with an arbitrary point **P**; on the plane. Note that we enter the two points as vectors and the normal vector as a rooted vector.

```
> P1:=<4,8,2>;P:=<x,y,z>;a:=RootedVector(root=P1,[4,-6,8]);
```

$$P1 := \begin{bmatrix} 4 \\ 8 \\ 2 \end{bmatrix}$$

$$P := \begin{bmatrix} x \\ y \\ z \end{bmatrix}$$

$$a := \begin{bmatrix} 4 \\ -6 \\ 8 \end{bmatrix}$$

We find an equation for the plane.

```
> plane:=DotProduct(a,P-P1)=0;
```

$$plane := 4\,x + 16 - 6\,y + 8\,z = 0$$

We plot the plane along with the normal vector and a vector in the plane, both vectors with tail at **P1**.

```
> p1:=implicitplot3d(plane,x=-10..10,y=-10..10,z=-10..10,style=
  patchnogrid):
  p2:=PlotVector(RootedVector(root=P1,[-8,-8,-2]),width=.4,
  head_width=.8,head_length=.8,scaling=constrained,color=red):
  p3:=PlotVector(a,width=.4,head_width=.8,head_length=.8,scaling=
  constrained,color=blue):
  p6:=textplot3d({[3,6,2,"v1"],[0,5,8,"v3"],[1.5,1,0,"w1"],[0,4,6,
  "w3"],[0,-1.5,1,"v5"],[0,-1,0,"w5"]},font=[TIMES,BOLD,14],color=
  black):
  display(p1,p2,p3);
```



We next find the intersection of the above plane with the plane -6x-5y+4z-3=0.

```
> plane2:=-6*x-5*y+4*z-3=0;
```

$$plane2 := -6\,x - 5\,y + 4\,z - 3 = 0$$

To find parametric equations for the intersection, we solve each equation for the same variable, here *x*.

```
> firstx:=solve(plane,x);
```

$$firstx := -4 + \frac{3}{2}\,y - 2\,z$$

```
> secondx:=solve(plane2,x);
```

$$secondx := -\frac{5}{6}y + \frac{2}{3}z - \frac{1}{2}$$

We then equate the two representations of the first variable and solve for a second variable, here y.

```
> y:=solve(firstx=secondx,y);
```

$$y := \frac{3}{2} + \frac{8}{7}z$$

This representation of the second variable is then substituted into either of the representations of the first variable, with the third variable, here z, being set equal to the parameter t.

```
> x1:=firstx;x2:=secondx;
```

$$x1 := -\frac{7}{4} - \frac{2}{7}z$$

$$x2 := -\frac{7}{4} - \frac{2}{7}z$$

We now have the parametric equations $x = -\frac{7}{4} - \frac{2}{7}t,\ y = \frac{3}{2} + \frac{8}{7}t,\ z = t$ for the line of intersection.

Before plotting, we need to reset two variables.

```
> y:='y';z:='z';
```

$$y := y$$

$$z := z$$

```
> p1:=implicitplot3d(plane,x=-10..10,y=-10..10,z=-10..10,style=
  patchnogrid,color=green):
  p2:=implicitplot3d(plane2,x=-10..10,y=-10..10,z=-10..10,style=
  patchnogrid,color=magenta):
  p3:=spacecurve([-7/4-2/7*t,3/2+8/7*t,t],t=-10..10,color=blue,
  thickness=4):
  display(p1,p2, p3);
```



10
−10
y
10
5
0

>